Oxford: smooth fit to log-odds ratios

Breslow and Clayton (1993) re-analyse 2 by 2 tables of cases (deaths from childhood cancer) and controls tabulated against maternal exposure to X-rays, one table for each of 120 combinations of age (0-9) and birth year (1944-1964). The data may be arranged to the following form.

Strata	Exposure: X-ray / total			
	Cases	*Controls*	*age*	*year - 1954*
1	3/28	0/28	9	-10
.....				
120	7/32	1/32	1	10

Their most complex model is equivalent to expressing the log(odds-ratio) ψ_i for the table in stratum *i* as

$$\log\psi_i = \alpha + \beta_1 \mathrm{year}_i + \beta_2(\mathrm{year}_i^2 - 22) + b_i$$

$$b_i \sim \mathrm{Normal}(0, \tau)$$

They use a quasi-likelihood approximation of the full hypergeometric likelihood obtained by conditioning on the margins of the tables.

We let r^0_i denote number of exposures among the n^0_i controls in stratum *i*, and r^1_i denote number of exposures for the n^1_i cases. The we assume

$$r^0_i \sim \mathrm{Binomial}(p^0_i, n^0_i)$$

$$r^1_i \sim \mathrm{Binomial}(p^1_i, n^1_i)$$

$$\mathrm{logit}(p^0_i) = \mu_i$$

$$\mathrm{logit}(p^1_i) = \mu_i + \log\psi_i$$

Assuming this model with independent vague priors for the μ_i's provides the correct conditional likelihood. The appropriate graph is shown below



BUGS *language for Oxford example:*

```
model
{
   for (i in 1 : K) {
      r0[i]  ~ dbin(p0[i], n0[i])
      r1[i] ~ dbin(p1[i], n1[i])
      logit(p0[i]) <- mu[i]
      logit(p1[i]) <- mu[i] + logPsi[i]
      logPsi[i]    <- alpha + beta1 * year[i] + beta2 * (year[i] * year[i] - 22) + b[i]
      b[i] ~ dnorm(0, tau)
      mu[i]  ~ dnorm(0.0, 1.0E-6)
   }
   alpha  ~ dnorm(0.0, 1.0E-6)
   beta1  ~ dnorm(0.0, 1.0E-6)
   beta2  ~ dnorm(0.0, 1.0E-6)
   tau    ~ dgamma(1.0E-3, 1.0E-3)
   sigma <- 1 / sqrt(tau)
}
```

Data (click to open)

Inits (click to open)

Results

A 1000 update burn in followed by a further 10000 updates gave the parameter estimates

	mean	**sd**	**MC_error**	**val2.5pc**	**median**	**val97.5pc**	**start**	**sample**
alpha	0.579	0.062	0.001545	0.4587	0.5793	0.7037	1001	10000
beta1	-0.04557	0.01553	3.929E-4	-0.07688	-0.0457	-0.01586	1001	10000
beta2	0.007041	0.003084	8.953E-5	0.001018	0.007004	0.01314	1001	10000
sigma	0.09697	0.06011	0.005036	0.02419	0.08059	0.2457	1001	10000

These estimates compare well with Breslow and Clayton (1993) PQL estimates of α = 0.566 +/- 0.070, β_1 = -0.469 +/- 0.0167, β_2 = 0.0071 +/- 0.0033, σ = 0.15 +/- 0.10.